# Tensor Omics: Team Coding Guidelines

August 14, 2026

## Contents

# 1 Compiler: Always Use the Latest gfortran

Always use `gfortran` 15 or newer. Newer compiler versions bring improved standard compliance, better auto-vectorization, and improved `do concurrent` support.

## 1.1 Using Docker (Recommended)

The project provides a pre-built Docker image with gfortran 15 and all required dependencies: no system-level changes are needed. Install Docker as explained for your operating system in the Docker documentation.

**Step 1:** Build the image from the `Dockerfile` in `misc/`:

```
cd misc
docker build -t arch-gfortran -f gfortran.docker .
cd ..
```

**Step 2:** Compile the project inside the container:

```
docker run -it -v $(pwd):/opt -w /opt arch-gfortran ./build.sh
```

## 1.2 Native Compilation

If you already have `gfortran` $\geq$ 15 installed, compile directly with `build.sh`. It compiles all files in `src/`, places compiled objects under `build/<compiler>/`, and creates a symbolic link to the resulting shared library (`libtensor-omics.so`) so that Python and R always find the same path.

**Arch Linux** (rolling release — always has the latest GCC):

```
sudo pacman -S gcc-fortran
```

**macOS** (via Homebrew):

```
brew install gcc
```

On Ubuntu/Debian, gfortran 15 is not available in the standard repositories. Use the Docker path instead.

**Default** — gfortran without optimisation flags:

```
./build.sh
```

**Maximum performance** — gfortran with optimisation flags:

```
./build.sh --max-performance
```

**Intel Fortran Compiler** — with maximum performance flags:

```
./build.sh --max-performance FC=ifx
```

## 1.3 Fortran Standard Reference

When in doubt about language semantics, consult the official Fortran standard draft:

https://j3-fortran.org/doc/year/26/26-007.pdf

# 2 File Extensions: `.F90` vs `.f90`

Fortran source files in this project use *two* extensions with a precise meaning:

- `.F90` (uppercase F): The file is passed through the C preprocessor before compilation. Use this extension whenever the file contains `#include`, `#define`, or any other preprocessor directive (e.g., `#include "macros.h"`).

- `.f90` (lowercase f): Plain Fortran, *no* preprocessor. Use this for files that contain no macros or includes.

**Rule:** If in doubt, use `.F90`. Do not mix files that use macros with the `.f90` extension, as gfortran will silently skip the preprocessor pass and produce cryptic compilation errors.

# 3 Code Formatting

## 3.1 Indentation

Use **4 spaces** per indentation level. Never use tabs. The project formatter is `fprettify`; run it before committing:

```
fprettify -i 4 -l 1000 <path-to-file>
```

The `-l 1000` flag sets a generous line-length limit so that `fprettify` does not introduce spurious line continuations in long argument lists.

## 3.2 One Dummy Argument Declaration Per Line

Each dummy argument (subroutine/function parameter) must be declared on its own line. This is required so that every argument can carry its own FORD documentation comment:

```
! CORRECT
integer(int32), intent(in) :: n_genes
  !! Total number of genes
real(real64), intent(in) :: expression_vectors(n_axes, n_genes)
  !! Gene expression matrix (n_axes x n_genes)

! WRONG: two arguments on one line cannot be individually documented
integer(int32), intent(in) :: n_genes, n_axes
```

For **local variables** this rule does not apply — grouping related locals on one line is acceptable. Local variables do not need FORD comments, but must still follow the naming conventions (meaningful, descriptive names).

# 4 Always Use `c_bool` for Logicals

Every logical that occupies storage — dummy arguments, locals, module variables, arrays — is declared `logical(c_bool)` from `iso_c_binding`, never the default kind.

```
use, intrinsic :: iso_c_binding, only: c_bool

logical(c_bool), intent(in) :: selection_mask(n_genes)
logical(c_bool) :: found
```

**Why.** A default `logical` is four bytes and `c_bool` is one, so a mask over genes costs a quarter as much — and it is the kind C, Python and R already use, so a mask crosses the boundary with no copy at all. Before this, every generated C wrapper declared an automatic array of default `logical` and converted elementwise on the way in and out.

**Two exceptions, both because there is no storage to save.** A `logical, parameter` is a named constant. A `logical function` result is a scalar, and a condition accepts any logical kind, so nothing forces it either way. Neither is worth the noise.

**A literal needs the kind too.** `.true.` is default kind, and argument association does not convert kinds, so passing it to a `logical(c_bool)` dummy is a compile error — write `.true._c_bool`. Assignment is fine and converts silently, which is why `mask = x > 0.0_real64` needs nothing special.

**Characters are not the same case.** There is no matching rule to use `character(kind=c_char)` everywhere: an implementation declares an ordinary `character(len=*)`, and `c_char` appears only where

a buffer really is a C buffer, which in hand-written code is the libzip interface and the view helper in `tox_conversions`. What an implementation *does* need to know is that a character argument arrives blank-padded to the buffer width rather than trimmed — see `codegen_guide.md`.

# 5 Always Use Double Precision

All real variables and constants must use the ISO Fortran double precision kind `real64` from `iso_fortran_env`. Never use the default `real` kind or literal constants without a kind suffix.

```
use, intrinsic :: iso_fortran_env, only: real64, int32

real(real64) :: x, y
integer(int32) :: n

! Compile-time constants MUST carry the _real64 suffix
x = 3.141592653589793_real64
y = 2.718281828459045_real64
```

Omitting the `_real64` suffix causes the literal to be evaluated as single precision before being promoted, silently losing bits of accuracy.

# 6 Variable and Argument Naming Conventions

Consistent naming makes the code readable without comments. The following conventions are **mandatory**.

## 6.1 Prefixes and Suffixes

| Pattern | Meaning | Example |
|---|---|---|
| `n_` | Size / count of a dimension | `n_genes`, `n_axes`, `n_families` |
| `i_` | Loop / iteration variable | `i_gene`, `i_axis`, `i_vec` |
| `_idx` | Calculated / derived index | `family_idx`, `out_idx` |
| `tmp_` | Work array (not input or output) | `tmp_perm`, `tmp_loess_y` |
| `_perm` | Permutation vector for `<name>` | `sort_perm`, `gene_perm` |
| `_impl` | Core implementation subroutine, and the module holding it | `k_means_assign_cluster_impl` |
| `_c` | C-interoperable wrapper (generated) | `compute_shift_vector_field_c` |
| `_expert` | Full-control wrapper with all work arrays exposed (generated) | `compute_family_scaling_expert` |
| `ierr` | Error code argument (always this name) | `integer(int32), intent(out) :: ierr` |

## 6.2 Mapping Arrays

Use a descriptive `x_to_y` pattern for mapping arrays:

```
integer(int32), intent(in) :: gene_to_fam(n_genes)
! gene_to_fam(i_gene) = family index of gene i_gene

integer(int32) :: family_idx
family_idx = gene_to_fam(i_gene)
```

## 6.3 Boolean / Selection Arrays

Use the `_mask` or `_selection` suffix for logical index arrays. Every logical that occupies storage is declared `logical(c_bool)` — one byte rather than four, and the kind the bindings already use, so a mask crosses to C, Python and R without being copied:

```
logical(c_bool), intent(in) :: vecs_selection_mask(n_vecs)
logical(c_bool), intent(in) :: axes_selection(n_axes)
```

# 7 Writing for the Code Generator

You write **one** procedure per computation: the implementation. The code generator reads it and writes everything around it — the validating and allocating entry points, the `bind(C)` wrapper, the Python and R bindings, their documentation, and the VS Code snippets. None of those may be edited by hand; a change belongs in the implementation or in its annotations. This section is the whole contract. `codegen_guide.md` in the repository root works through it case by case, with an example for each.

## 7.1 The Implementation

- Name it `foo_impl`, in a module whose name ends `_impl`. That module name is what triggers generation, and the generated module is written to `src/generated/` mirroring the implementation's own path.
- It contains the algorithm and nothing else: it **never validates input** and **never allocates** — nor may anything else in an `_impl` module. It may still report a genuine *runtime* error through `ierr` (a division by zero, a failed external fit); validation is not its job.
- Accept *everything* as explicit arguments: inputs, outputs, and any work arrays. Declare it `pure` unless the algorithm genuinely cannot be. Purity is derived, not declared, for everything above it.
- An argument named `tmp_<name>` is a **work array**: the allocating entry point allocates it and drops it from the signature the caller sees.
- An argument named `<base>_perm` is a **sorting permutation** for `<base>`: the allocating entry point initialises and heapsorts it for you.

## 7.2 Documentation Macros (`DM_` prefix)

Every constraint the generated layers need is written as a `DM_` macro in the argument's own `!!` comment. They are documentation macros: they expand to the English sentence that ends up in the Fortran, C, Python and R documentation, *and* the generator reads them to emit the matching validation. One statement, one place, no way for the check and the prose to disagree.

| Macro | Purpose |
|---|---|
| `DM_MIN(EXPR)` | Lower bound for this argument |
| `DM_MAX(EXPR)` | Upper bound for this argument |
| `DM_SENTINEL(EXPR)` | One extra value accepted outside the bounds |
| `DM_ALLOW_NAN` | Permit NaN (rejected by default for every real argument) |
| `DM_ALLOW_INFINITE` | Permit $\pm\infty$ (rejected by default for every real argument) |
| `DM_DEFAULT(VAL)` | Default for an `optional` argument |
| `DM_RESULT_SIZE_IS(ARG)` | Only the first `ARG` elements of this output are filled |
| `DM_OUTPUT_FROM(...)` | Compute this argument by calling a sizing routine |
| `DM_REQUIRED_IF_MODE(...)` | This optional argument is required in one mode |
| `DM_PROLOGUE(PROC, MOD)` | Run `PROC` before the implementation, and let it short-circuit |

**Finiteness is the default contract**: every real argument is checked for NaN and infinity unless it opts out. A bound may be an expression, including another argument (`DM_MAX(n_array)`) or an exclusive bound (`DM_MIN(above(0.0_real64))`).

```
pure subroutine quantile_normalization_impl(expr, n_genes, n_replicates, &
    normalized_expr, rank_means, tmp_col, tmp_perm)
  integer(int32), intent(in) :: n_genes
```

```
    !! Number of genes (rows)
  integer(int32), intent(in) :: n_replicates
    !! Number of replicates per gene
  real(real64), intent(in)  :: expr(n_replicates, n_genes)
    !! Gene expression matrix
    !! DM_ALLOW_NAN
    !! DM_ALLOW_INFINITE
  real(real64), intent(out) :: normalized_expr(n_replicates, n_genes)
    !! Normalized `expr`
  real(real64), intent(in)  :: rank_means(n_replicates)
    !! The mean of each rank across tissues
  real(real64), intent(out) :: tmp_col(n_genes)      ! work array
  integer(int32), intent(out) :: tmp_perm(n_genes)   ! permutation work array
  ! ... pure implementation: no ierr, no validation, no allocation ...
end subroutine
```

## 7.3 What Comes Out

Two Fortran entry points, and the bindings around them:

- `foo` — validates, allocates the work arrays, sorts the permutations, and calls the implementation. **The entry point a caller reaches for first**, and it takes `ierr` last, after the optional arguments.
- `foo_expert` — validates and calls the implementation with what you supply, allocating and preparing nothing. Emitted only where there is something to take over; where there is not, there is a single wrapper and it is called `foo`.
- `foo_c`, the Python `ctypes` module, the R wrapper and its C `.Call` shim, and the snippets — all from the same source, with the documentation rendered per language.

**Your doc comments are the published API.** An implementation's `!>` and `!!` blocks become the Fortran, C, Python and R documentation verbatim, so write them for a reader of those languages rather than as notes to yourself. A `[[link]]` that names nothing is a build error.

## 7.4 The Workflow

When implementing new scientific functionality, complete these steps *in order* and include all of them in the same Merge Request:

1. **Implementation.** Write the annotated `foo_impl` as above. This is the only layer you write.
2. **Generate.** Run `python helper/generate_code.py`; every `./build.sh` does it for you unless `--skip-code-generatio` is passed. Commit the regenerated files with your change. If a published procedure was renamed, also run `--target snippets`, which is not one of the default targets.
3. **Tests.** Fortran unit tests for numerical correctness, plus Python and R tests — see the Testing section. These stay hand-written.

## 7.5 When Not to Write a Procedure At All

Only add functionality that is genuinely absent in Python and R. Do not wrap:

- Sorting routines (Python `numpy.sort`, R `order()` already exist).
- Generic I/O or string utilities.

### 7.6 If the Generator Cannot Express It

The annotation vocabulary is deliberately about language structure, never about domain concepts. If a procedure's contract cannot be expressed in it, that is a sign the *procedure* needs rethinking rather than the generator needing a special case. Raise it rather than working around it.

## 8 Error Handling

All errors are communicated through a single integer output argument `ierr`. The `tox_errors` module defines all error codes and the functions to set and check them.

### 8.1 Error Code Categories

| Range | Category | Example constant |
|---|---|---|
| 0 | Success | `ERR_OK` |
| 1xx | I/O and file errors | `ERR_FILE_OPEN`, `ERR_READ_DATA` |
| 2xx | Input validation | `ERR_INVALID_INPUT`, `ERR_DIM_MISMATCH`, `ERR_NAN_INF` |
| 3xx | Memory | `ERR_ALLOC_FAIL`, `ERR_POINTER_NULL` |
| 5xxx | Fortran runtime | `ERR_UNIT_NOT_CONNECTED` |
| 9xxx | Internal / unknown | `ERR_INTERNAL`, `ERR_UNKNOWN` |

### 8.2 Always Initialise `ierr` First

Call `set_ok(ierr)` at the start of every subroutine that uses `ierr`:

```
use tox_errors, only: set_ok, set_err, set_err_once, is_err, ERR_INVALID_INPUT

call set_ok(ierr)    ! ierr = ERR_OK = 0
```

### 8.3 Return Early on Error

Check `ierr` after every operation that can fail, and return immediately:

```
call validate_dimension_size(n_genes, ierr)
call validate_all_in_range_real(data, n_genes, ierr)
if (is_err(ierr)) return
```

### 8.4 Reporting the Failing Argument (`arg_pos`)

The generated entry points do all input validation, and number the argument positions for you. On the rare occasion that something has to be validated by hand, use the `tox_errors` validators rather than an ad-hoc check, and pass `arg_pos`:

```
call validate_dimension_size(n_genes, ierr, arg_pos=1_int32)
call validate_in_range_real(span, ierr, min=0.0_real64, max=1.0_real64, arg_pos=6
    _int32)
call validate_all_in_range_real(data_points, size(data_points, kind=int32), ierr,
    arg_pos=2_int32)
```

`arg_pos` must match the argument's position in the public (non-`_impl`) subroutine that callers actually invoke, never its position in whichever internal routine happens to run the check — Python and R index their argument-name tuple with it.

When the valid range of an argument is itself defined by `CM_` macros (Section 12.2), pass those macros as the bounds, so the reported range and the validated range can never disagree.

### 8.5 `set_err` vs `set_err_once`

- `set_err(ierr, code)`: *Unconditionally* sets `ierr` to `code`, overwriting any error already stored. Use it only when `ierr` is known to be `ERR_OK` (e.g. the first check after `set_ok`), or when you deliberately want the new code to take precedence.
- `set_err_once(ierr, code)`: Sets `ierr` to `code` *only if `ierr` is currently `ERR_OK`*, preserving the first error encountered. This is the safe default when chaining multiple fallible calls, and is exactly what the `validate_*` routines use internally.

### 8.6 Never Silent Errors

Do not ignore `ierr`. If a callee sets an error, the caller must either propagate it or handle it explicitly. Every downstream caller that receives an `ierr` must check it.

### 8.7 Avoid GOTO

**Never use `GOTO`.** It is prohibited in all new code for the following reasons:

- It creates non-local control flow that is impossible to follow, making error paths untraceable.
- It defeats the structured error model built around `ierr` and early returns.
- It prevents the compiler from performing flow analysis, blocking optimizations and `pure` qualification.
- It is incompatible with `do concurrent`, which requires structured concurrency regions.

**How to replace `GOTO` correctly:**

| Pattern to replace | Correct Fortran replacement |
|---|---|
| Forward jump to error label | `call set_err(ierr, ...); return` |
| Loop exit (`GOTO after loop`) | `exit` or `cycle` inside a `do` loop |
| Conditional skip block | `if (...) then ... end if` |
| Multi-level exit | Refactor into a subroutine with early `return` |

```
! WRONG
if (n <= 0) goto 99
call compute(n)
99 continue

! CORRECT
if (n <= 0) then
  call set_err(ierr, ERR_INVALID_INPUT)
  return
end if
call compute(n)
```

## 9 Use of `intent` and `pure`

### 9.1 Mandatory `intent`

Every argument in every subroutine and function must carry an `intent` attribute:

- `intent(in)`: The routine only reads this argument.
- `intent(out)`: The routine completely overwrites this argument.
- `intent(inout)`: The routine both reads and modifies the argument, and the incoming value matters.

Omitting `intent` disables aliasing checks and is forbidden.

### 9.2 Mandatory `pure` for Core Routines

All *helper* and core subroutines must be declared `pure`. A `pure` procedure:

- Has no side effects (no I/O, no global state changes).
- Is eligible for auto-vectorization and threading by the compiler.
- Allows the compiler to schedule calls more aggressively.

Remove `pure` only when truly necessary (e.g., I/O, or an impure external library). Allocation is not a reason: `allocate(..., stat=)` is legal in a pure procedure.

### 9.3 Optional Arguments

Fortran does not support default values in argument lists. Use the `M_DEFAULT_VAL` macro to assign defaults after a `PRESENT` check:

```
#include "macros.h"
subroutine my_routine(data, n, span, degree)
  real(real64), intent(in), optional :: span
  integer(int32), intent(in), optional :: degree
  real(real64) :: actual_span
  integer(int32) :: actual_degree

  M_DEFAULT_VAL(span, actual_span, 0.7_real64)
  M_DEFAULT_VAL(degree, actual_degree, 2_int32)
```

**Never** use optional arguments in C wrappers; `bind(C)` is incompatible with them.

## 10 Parallelisation: `do concurrent`

Prefer `do concurrent` over plain `do` loops whenever the iterations are independent. This is the primary parallelisation mechanism in this project. Do not use OpenMP pragma loops — let the compiler exploit `do concurrent` through its own vectorisation and threading back-ends.

### 10.1 Locality Specifications

Every variable used inside a `do concurrent` loop **must** carry an explicit locality specification. Do not rely on the compiler's assumption — leaving it implicit makes the intent ambiguous and can produce incorrect results with aggressive optimisers.

| Specification | Meaning |
|---|---|
| `shared(...vars)` | All iterations read/write the *same* instance of the variable. Use for arrays being filled element-wise or for read-only scalars. |
| `local(...vars)` | Each iteration gets its own *private* copy, uninitialised at entry. Use for loop-local temporaries. |
| `local_init(...vars)` | Like `local`, but each copy is initialised to the value the variable had before the loop. |
| `reduce(op:...vars)` | Each iteration contributes to a scalar accumulator via `op`. The final result is the reduction of all per-iteration contributions. |

Supported `reduce` operators: `+`, `*`, `max`, `min`, `iand`, `ieor`, `ior`, `.and.`, `.or.`, `.eqv.`, `.neqv.`

```
! Element-wise: array filled in parallel, scalar read-only
do concurrent (i_dim = 1:n_dims) shared(vector, vector_norm)
  vector(i_dim) = vector(i_dim) / vector_norm
end do

! Reduction: accumulate a sum over independent iterations
means_sum = 0.0_real64
do concurrent (i_gene = 1:n_genes) shared(gene_means) reduce(+:means_sum)
  means_sum = means_sum + gene_means(i_gene)
end do

! Local temporary: each iteration owns its own scratch variable
do concurrent (i_vec = 1:n_vectors) shared(mat, norms) local(tmp_norm)
  tmp_norm = norm2(mat(:, i_vec))
  norms(i_vec) = tmp_norm
end do
```

**When to use `do concurrent`:**

- Element-wise array operations.
- Independent distance or dot-product computations over gene/axis indices.
- Any loop where iteration order does not affect correctness.

**When NOT to use `do concurrent`:**

- Loops with data-dependent control flow across iterations (e.g., early-exit on error found inside a loop — in that case, validate before the loop).
- Sequential algorithms like merge steps of merge-sort.

# 11 Memory Model: No Copies, No Allocations in Cores

## 11.1 No Allocations in Core Routines

**Nothing in an `_impl` module may allocate memory**, helpers included. Allocation is performed exclusively in the generated allocating entry point, and the generator enforces this. This rule:

- Keeps core routines `pure` and deterministic.
- Allows callers to reuse pre-allocated work arrays across multiple calls, avoiding repeated `malloc`/`free` cycles over large datasets.

## 11.2 No Defensive Copies

Avoid creating copies of input arrays. For large omics datasets (tens of thousands of genes, hundreds of tissues), an unexpected copy easily exceeds available memory. Design algorithms to work on slices and views of the original data.

## 11.3 Logical Masks

A logical mask is a `logical` array with the same shape as the data it describes: `.true.` marks an element to include, `.false.` marks it to exclude. Passing a mask alongside the full array lets a routine include or exclude elements in place — with a plain `if (mask(...))` check — instead of first allocating a smaller array and copying the kept elements into it. This saves both the allocation and the copy, which matters at big scale.

Prefer adding a mask argument over asking the caller to pre-filter their data into a separate array. In `tox_relative_axis_plane_tools.F90`, `omics_vector_RAP_projection` takes the full `vecs` matrix together with a `vecs_selection_mask`/`axes_selection_mask`, and only writes the selected entries into the (exactly-sized) output — the input matrix itself is never copied or filtered:

```
logical, dimension(n_vecs), intent(in) :: vecs_selection_mask
    !! `.true.` for vectors where projection is to be computed
...
do i_vec = 1, n_vecs
    if (vecs_selection_mask(i_vec)) then
        ...
        if (axes_selection_mask(i_axis)) then
            projections(i_axis_proj, i_vec_proj) = vecs(i_axis, i_vec)
            ...
```

Similarly, in `tox_data_integration_jsd_impl.F90`, an optional `neighbor_mask` lets `build_residual_histograms_impl` skip specific neighbors while iterating the original, unfiltered array:

```
logical, dimension(n_neighbors, n_points), intent(in), optional :: neighbor_mask
    !! Optional mask to exclude specific neighbors (e.g. for family-wise analysis)
...
filter_neighbors = present(neighbor_mask)
...
if (filter_neighbors) then
    if (.not. neighbor_mask(i_neighbor, i_point)) cycle
end if
```

No subset of `neighborhood_residuals` is ever built — the mask only decides which iterations contribute.

When a smaller, compacted array genuinely has to be produced (e.g. to hand a reduced gene list downstream), use `count(mask)` to size the allocation exactly once, rather than over-allocating or growing it as elements are found, as in `tox_data_tools.F90`:

```
n_genes_kept = count(mask)
allocate(temp_gene_ids(n_genes_kept), stat=ios)
...
do i = 1, n_genes_total
    if (mask(i)) then
        temp_gene_ids(j) = gene_ids(i)
        ...
        j = j + 1
    end if
end do
```

Use masks wherever a routine only needs to act on part of its input: they are cheaper than copying, and they keep the calling convention uniform (same array, same shape, one extra `logical` argument) whether or not filtering is requested.

## 11.4 Work Arrays (`tmp_`)

Work arrays must be pre-allocated by the caller and passed as explicit arguments to core routines. They follow the `tmp_` prefix convention.

```
! tmp_perm holds the reusable sort permutation
integer(int32), intent(out) :: tmp_perm(n_genes)
real(real64),   intent(out) :: tmp_loess_y(n_genes)
```

### 11.5 Reusing Output Buffers as Work Arrays (Pointer Reshaping)

When a routine has already been handed a large enough output (or other pre-allocated) buffer, reuse it as scratch space via Fortran pointer bounds-remapping instead of allocating a separate temporary. This is a zero-copy *reshape* of contiguous storage: the pointer sees the same memory under new bounds.

```
! Reinterpret a contiguous buffer as a 2-D work view (no copy, no allocation)
real(real64), dimension(:, :), pointer :: work_view
work_view(1:n_genes, 1:n_tissues) => output_buffer
! A column slice can then serve as a work pointer
tmp_col_ptr => work_view(:, 1)
```

**This reshapes; it does not transpose.** Bounds-remapping reinterprets the linear, column-major storage under the new shape: `work_view(i, j)` is element `(j-1)*n_genes + i` of `output_buffer`, *not* `output_buffer(j, i)`. A genuine transpose reorders elements and therefore requires moving data (e.g. `transpose` into a separate buffer); a pointer view can never transpose. The target must be contiguous.

Only reuse a buffer as scratch when its final output value is not needed until after the scratch use has finished — otherwise the scratch writes corrupt the output.

## 12 Macro Usage (`src/macros.h`)

All source files that use macros must start with:

```
#include <src/macros.h>
```

**Never write a bare `implicit none`**: every module uses `M_IMPLICIT_NONE` instead. It expands to `implicit none (type, external)`, the F2018 form, which additionally requires an explicit interface for every external procedure — so calling one that the compiler has never seen becomes a compile error at the call site rather than a link-time surprise or, worse, a silently wrong ABI.

### 12.1 Available Macros

| Macro | Purpose |
|---|---|
| `M_IMPLICIT_NONE` | `implicit none (type, external)`; use it in every module |
| `M_USE_NULL_VALIDATION` | Imports modules needed for `c_loc`/null checks in C wrappers |
| `M_CHECK_IERR_NON_NULL` | Silently returns if `ierr` pointer is null (always first check) |
| `M_CHECK_NON_NULL(ARG)` | Sets `ierr` to `ERR_POINTER_NULL` and returns if `ARG` pointer is null |
| `M_DEFAULT_VAL(OPT,LOC,DEF)` | Assigns `OPT` to `LOC` if present, else assigns `DEF` |
| `M_ALLOCATE(DECL)` | Allocates `DECL`; sets `ERR_ALLOC_FAIL` and returns on failure |
| `M_NAN` | IEEE quiet NaN (double) |
| `M_NEG_INF` | IEEE negative infinity (double) |
| `M_POS_INF` | IEEE positive infinity (double) |

### 12.2 Module-Local Constants via `#define` (`CM_` prefix)

Global, reusable macros live in `macros.h` with the `M_` prefix. A constant that only makes sense within a single module does not belong there — adding it would clutter a file meant for cross-module reuse. Instead, define it locally, immediately after the `#include` line, using the `CM_` prefix ("Custom Macro"), and keep it next to the module's other constants (its `parameter` declarations):

```
#define CM_METHOD_AVERAGE 0
#define CM_METHOD_WEIGHTED 1
#define CM_METHOD_WARD 2

integer(int32), parameter :: METHOD_AVERAGE  = CM_METHOD_AVERAGE
integer(int32), parameter :: METHOD_WEIGHTED = CM_METHOD_WEIGHTED
integer(int32), parameter :: METHOD_WARD     = CM_METHOD_WARD
```

**Keeping FORD documentation in sync.** FORD does not resolve Fortran `parameter` values, but it does run the C preprocessor before generating docs, so a `CM_` macro referenced inside a doc comment renders as-is. This lets a mode/method table stay anchored to the code instead of to a literal that can drift:

```
!! |       Method       |           Value          |
!! |--------------------|--------------------------|
!! | Average / UPGMA    |   CM_METHOD_AVERAGE      |
!! | Weighted / WPGMA   |   CM_METHOD_WEIGHTED     |
!! |       Ward         |   CM_METHOD_WARD         |
```

**Use the macro in validation too.** When a `CM_` macro defines the valid values or bounds of an argument, pass the macro itself to the corresponding `validate_*` call instead of repeating the literal:

```
call validate_in_range_int(method, ierr, min=CM_METHOD_AVERAGE, max=CM_METHOD_WARD,
    arg_pos=7_int32)
```

This keeps the default/bound, the validation, and the documentation all traceable to one definition — if a mode is added or reordered, only the `#define` block needs to change.

# 13 Derived Types (`type`): Use with Care

Derived types (`type :: ...`) are a powerful Fortran feature but must be used with restraint in this project:

- **Prohibited** in any subroutine or function that has a C wrapper or is otherwise called from Python or R. C, Python `ctypes`, and R's C interface cannot represent Fortran derived types; passing them across the language boundary is undefined behaviour.
- **Permitted** for pure-Fortran data structures where the extra abstraction genuinely improves readability or correctness (e.g., hash maps, JSON serialisation helpers).
- **Avoid by default** in the main TOX scientific routines, which operate exclusively on plain arrays and matrices.

If you need a compound structure that must cross the language boundary, decompose it into flat arrays and pass its components individually.

# 14 The `f42_utils` Utility Module

`src/f42/utils/` contains general-purpose helper functions shared across the project, split by topic into `f42_math_impl`, `f42_sort_impl`, `f42_random_impl`, `f42_vector_impl` and `f42_stats_impl`: `mean`, `std_dev`, `norm`, `clamp`, `sort_array`, `angle_between`, `is_close`, `logx`, `compute_edf`, etc. `f42_utils_impl` re-exports all five, so one `use` reaches everything.

**Rule:** Before implementing a new helper function anywhere else in the codebase, check whether the family already provides it. If you write a function that is genuinely reusable across modules, add it there rather than creating a module-local copy. This prevents duplication and keeps utility logic in one place. Import from the module that *defines* the helper where you know it, and from the parent where you do not:

```
use f42_math_impl, only: norm, is_close, logx, mean, std_dev
use f42_sort_impl, only: sort_array
```

# 15 The `f42_safeguard` Module

Any module that crosses the C boundary and treats `c_int` as `int32` or `c_double` as `real64` must `use f42_safeguard`. It contains compile-time assertions that those kind constants really are equivalent on the target platform:

```
use f42_safeguard   ! guarantees c_int == int32, c_double == real64,
    c_double_complex == real64
```

The generated `bind(C)` wrappers carry this themselves, so in practice the rule bites only where a module declares a `bind(C)` interface by hand.

If these kinds do not match, the module fails to compile with a descriptive error, preventing silent ABI (Application Binary Interface) mismatch bugs. The ABI is the low-level contract between compiled code: it specifies how function arguments are laid out in memory, what sizes the types occupy, and how values are returned. When the Fortran kind for `real64` (8 bytes) does not match the C kind for `c_double` on a given platform, data passed through a C wrapper would be misinterpreted silently, producing incorrect results without any runtime error.

# 16 FORD Documentation

**FORD** (Fortran Documentation Generator) produces HTML documentation directly from structured comments. All public procedures and modules must be documented.

## 16.1 Comment Syntax

- `!>` — Starts the description of a module, subroutine, or function.
- `!|` — Continues or extends a description (e.g. a second line after `!>`); also used to document an argument when placed **before** the declaration.
- `!!` — Trail comment placed **after** the declaration (on the following line, indented). This is the preferred way to document individual arguments in this codebase.

In summary: `!|` before, `!!` after. Both are valid FORD syntax; the template below uses `!!` consistently.

## 16.2 Module and Subroutine Template

```
!> Module for computing expression centroids of gene families.
!| Contains the core scientific kernel. C and language wrappers are defined
!| outside the module for compatibility.
module tox_gene_centroids
  use, intrinsic :: iso_fortran_env, only: real64, int32
  M_IMPLICIT_NONE
contains

  !> Computes the element-wise mean for a given set of gene expression vectors.
  !| Returns a single centroid vector of length n_axes.
  pure subroutine mean_vector(expression_vectors, n_axes, n_genes, &
      gene_indices, n_selected_genes, centroid, ierr)
    integer(int32), intent(in) :: n_axes
      !! Number of axes (dimensions)
    integer(int32), intent(in) :: n_genes
      !! Total number of genes in the expression matrix
    real(real64), intent(in) :: expression_vectors(n_axes, n_genes)
      !! Gene expression matrix (n_axes x n_genes)
    integer(int32), intent(in) :: n_selected_genes
      !! Number of genes to include in the centroid
```

```
    integer(int32), intent(in) :: gene_indices(n_selected_genes)
      !! Column indices of selected genes in expression_vectors
    real(real64), intent(out) :: centroid(n_axes)
      !! Output centroid vector (n_axes)
    integer(int32), intent(out) :: ierr
      !! Error code: 0 = success
    ! ... implementation ...
  end subroutine mean_vector

end module tox_gene_centroids
```

Note the use of `!!` (two exclamation marks) to place the documentation comment on the line *after* the declaration. This is the FORD convention used throughout the codebase.

### 16.3 Generating Documentation

Documentation is generated automatically by the CI/CD workflow on every merge to the main branch. **You do not need to run FORD manually.** However, if you want to preview the output locally before submitting an MR:

```
ford ford.yml         # run from project root
open doc/index.html   # review in browser
```

## 17 Testing

The project provides a structured test suite framework. See the test directory README for full details on how to add and run tests.

### 17.1 Where to Test What

- **Algorithm correctness**: Fortran unit tests only. The Fortran test suite is the single source of truth for numerical correctness.
- **Python / R wrappers**: Test only that the function can be called, that the return value is of the correct type and shape, and that reasonable input produces a non-error result.

The wrappers themselves are generated, but **their tests are not** — a generated binding is exactly the place where a mistake is uniform and invisible, so every published procedure needs a Python and an R test that actually calls it. Run everything with `./run_all_tests.sh`.

### 17.2 Test Naming Convention

Each test file is named `mod_test_<module_name>.{f90|F90|py|R}`. Each test function is named `test_<description>`.

## 18 Merge Request and Collaboration Conventions

- **Do not merge your own MR.** Always request at least one reviewer and wait for approval before merging.
- **Always assign experienced developers as reviewers.** Add at least Franz or Aaron as a reviewer on every MR. They have deep knowledge of the codebase and can catch issues early.
- **Do not close review threads yourself.** Only the person who opened a thread (the reviewer) should close it after it has been resolved to their satisfaction.

- **Resolve conflicts before requesting review.** Rebase onto the target branch and fix conflicts yourself.
- **Keep MRs focused.** One logical change per MR. Large refactors should be discussed in an issue before starting the MR.
- **Link to the relevant issue** in the MR description.
- **Update FORD documentation** and tests as part of the same MR as the implementation.
- **Check the CI/CD workflow.** After pushing or updating an MR, always verify that the automated workflow (tests, build, documentation generation) passes. If it fails, *fix it yourself* — do not wait to be told. A failing pipeline blocks the merge and means the code is not yet ready for review.
- **Use GitHub, not GitLab.** Since May 2026 the project is developed exclusively on GitHub. Do not open issues, branches, or MRs on the old GitLab repository — those will not be seen or merged.

# 19 Getting Help: The ask-for-help Channel

The ask-for-help channel is the right place for **any question** — not just programming. Whether it is about scientific context, workflow setup, naming decisions, tooling, or general project conventions, there will always be someone available to help. Do not hesitate to ask.

For *technical problems* (especially compilation errors), do not spend more than 30 minutes debugging alone. Post in the channel with:

1. The exact compiler command and full error message (copy-paste, do not paraphrase).
2. The minimal code snippet that triggers the error.
3. What you have already tried.

Many gfortran errors for `bind(C)`, `pure`, or kind-mismatch issues are non-obvious; the team has likely encountered the same problem before.

# 20 Integrating External Fortran Code from Netlib

Netlib is the canonical archive for legacy scientific Fortran (BLAS, LINPACK, LOESS, etc.). This section explains how to extract, compile, and link Netlib packages so they can be added to `external/`.

## 20.1 Shell Archives (shar)

Netlib sometimes distributes packages as **Shell Archives** (`.shar`): self-extracting shell scripts that reconstruct source files when executed. Each line of actual source is prefixed with a hyphen; the script uses `sed` to strip those prefixes during extraction and write the real files. This format dates to the 1980s and was also a safety measure against accidental execution of embedded shell commands.

Do not copy-paste the text manually. Save it to a file and run it:

```
wget https://netlib.org/a/loess
sh loess          # extracts README, loessf.f, lowesf.f, lowescp.f, ...
```

Perform extraction in a **temporary directory outside the project**. Once compilation is confirmed working, move the relevant files into `external/`.

### 20.2 Checking Dependencies

After extraction, read the `README`. Many Netlib packages omit routines they assumed were already present on the target system. LOESS, for example, depends on LINPACK routines (`dqrsl`, `dsvdc`) that are not included in the distributed archive. Missing symbols cause linker errors such as:

```
undefined reference to 'dqrsl'
undefined reference to 'dsvdc'
```

Fetch the missing `.f` files from Netlib and add them to the compilation.

### 20.3 Compiling Legacy Fortran Sources

Old Fortran (fixed-form, F77) requires special flags:

```
gfortran -O2 -std=legacy -ffixed-line-length-none \
         -fallow-argument-mismatch -fPIC -w \
         -c file1.f file2.f file3.f
```

| Flag | Purpose |
|---|---|
| `-O2` | Standard optimisations. |
| `-std=legacy` | Accepts F77 constructs no longer valid in modern standards. |
| `-ffixed-line-length-none` | Removes the 72-character column limit of fixed-form source. |
| `-fallow-argument-mismatch` | Suppresses fatal errors from implicit-interface argument-type mismatches common in old code. |
| `-fPIC` | Position-independent code; required when object files will be linked into a shared library (`.so`). |
| `-w` | Suppresses warnings. Use it only after the initial integration is stable; omit it while diagnosing problems. |
| `-c` | Compile only (produce `.o` files); do not link. |

### 20.4 Linking

To build a **static library** for inclusion in Tensor-Omics:

```
ar rcs libname.a *.o
```

To build a standalone **executable** for testing:

```
gfortran -o program_name file1.o file2.o file3.o
```

If `undefined reference` errors appear at link time, a dependency is missing. Classic culprits in Netlib packages are additional routines from **BLAS**, **LINPACK**, or utility functions such as `d1mach` — fetch and compile those as well.

## 21 Further Reading

This guide states the conventions. Four documents in the repository go deeper, and each is kept in step with the code:

- `codegen_guide.md` (repository root) — the author's guide to the code generator, case by case: ranges, masks, work arrays, permutations, sizing routines, modes and mode splits, prologues, runtime errors, split families. Every case has a real snippet and shows what comes out. **Read this before adding a procedure.**
- `helper/codegen/README.md` — what the generator produces and how to run it, including what it refuses and why.

- `helper/codegen/emit/README.md` — the language layers: every question the Python and R bindings had to answer (array layout, ownership of `intent(inout)`, null-as-absent, missing values, derived arguments, decoding `ierr`) and what each chose. Written so a third language decides rows instead of rediscovering them.
- `helper/codegen/design/` — the design of record, including every rejected alternative.

# 22 Quick-Reference Checklist

Use this checklist when reviewing or submitting any Fortran code contribution:

- ☐ File extension is `.F90` (macros used) or `.f90` (no macros).
- ☐ `#include <src/macros.h>` is present, and the module declares `M_IMPLICIT_NONE` rather than a bare `implicit none`.
- ☐ All `real` variables use `real(real64)`; all constants carry `_real64`.
- ☐ Every argument has `intent(in|out|inout)`.
- ☐ The implementation is named `foo_impl`, in a module ending `_impl`, and is declared `pure`.
- ☐ `do concurrent` is used for independent loops.
- ☐ No `GOTO` anywhere.
- ☐ The implementation validates nothing and allocates nothing; where it does report a runtime error, `ierr` is initialised with `set_ok(ierr)` and every failure path returns immediately.
- ☐ Every constraint is stated as a `DM_` macro on the argument it constrains, and every work array is named `tmp_`.
- ☐ No unexplained copies of large arrays.
- ☐ No derived types in anything that reaches C, Python or R.
- ☐ All utility functions are in the `f42_utils` family (or re-use existing ones).
- ☐ FORD comments (`!>`, `!|`, `!!`) are present for the module and every public procedure, written as API prose — they are what Python and R users read.
- ☐ The generator was run and its output committed (`./build.sh`, plus `--target snippets` if a published name changed).
- ☐ Fortran tests cover the new logic.
- ☐ Python/R tests verify call-ability and return type.
- ☐ MR does not self-merge and review threads are not self-closed.